

```

! RE-STORE "Skeleton.prg"
!
!*****
! "Skeleton program #1" for Tech Soft's MS Office function library
! Can be used as a framework for your own applications.
!
! (c) Tech Soft GmbH 1999-2007          http://www.TechSoft.de
! Author:          Sven Henze
! Date:           03-Sep-2001
! Last change:    26-Jan-2007
!*****
!
!
INTEGER Verbose=1          ! switch on/off verbose mode (0=off)
!
! Load the high level function library if not already loaded
IF NOT INMEM("Dde_lib_start") THEN LOADSUB ALL FROM "htbde.lib"
!
!
! Now we call the custom SUB where the Office functions will be executed
! If you need more calling parameters either include them in the SUB definition
! or use COM blocks
!
CALL Dde_mainprog(Verbose)
!
! cleaning up SUB programs (in order to make sure that the most recent versions are loaded every time)
DELSUB Dde_lib_start TO END
!
END
!
SUB Dde_mainprog(OPTIONAL INTEGER Verbose)
!
!*****
! *
! * MS Office Interface for HTBasic: Skeleton program
! *
! * Version : 2.10.27
! * Author  : Sven Henze, Tech Soft GmbH
! * Created : 03-Sep-2001
! * Updated : 26-Jan-2007
! *
! * Copyright (c) 1999-2007 Tech Soft GmbH. All Rights Reserved.
! *                               Email: HTB@TechSoft.de
! *                               http://www.TechSoft.de/mso
! *
!*****
!
! Description:
! =====
!
! The entry point for your own DDE commands can be found under the label
! named "Myown_dde_cmds". You can include any DDE code here using the
! pre-allocated system variables (DDE channel numbers, application name etc.).
! One example is given, it returns all available "Sysitems" of the application.
!
! There are some special predefined templates for Excel, Word and Access.
! Please select the appropriate value for the variable "App$" (see below)
! in order to select the desired application. You can easily adjust this
! skeleton program for other applications.
!
! Please note:
! This program was written and optimized for HTBasic 8.3 and higher.
! If you use an older version of HTBasic it might be necessary to remove
! or change some commands. The example does *not* work with HTBasic 7.x due
! to some problems in the CSUB interface within HTBasic 7.x.
!
! Note for MS Office 2007 users:
! A question dialog will be displayed where you will be asked if you are
! using MS Office 2007. You can suppress this question dialog by commenting
! out the line with the label "Office2007:"
!
!
! define some variables
!
DIM Doc_template$(1:10)[255],Doc_file$(255),App_exe$(255),Topic$(255),Tmp$(8192),Htbvers$(80),Os$(255),Question$(1024)
INTEGER Printer,Btn,Sys_chn(1:5),Doc_chn(1:10),Result(1:10),App_quit,Language,Printlen,Dummy,Mso2007
DIM Ddecmd$(255),Info$(2048),Bookmark$(128),Doc_password$(128),Sel$(255),Struct_file$(255)
DIM G$(5),X$(5),Hl$(5),Hx$(5),App$(255),CrLf$(8),Wpwinvers$(20),Ul$(10),Newfile$(5),Openopt$(20)
DIM Wordsuffix$(20),Excelsuffix$(20),Accesssuffix$(20) ! file type suffix for MS Office applications
!
G$=CHR$(19) ! character code for inverse printing
X$=CHR$(16) ! character code for normal printing
Hl$=CHR$(30) ! character code for "highlighted" printing (different color)
Hx$=CHR$(24) ! character code for "normal" printing (color)
Ul$=CHR$(22) ! character code for underlined printing
CrLf$=CHR$(10)&CHR$(13)
!
ON ERROR GOSUB Dde_err_handler ! Global error handler
!
!
INTEGER Verb_
Verb_=0
IF NPAR>=1 THEN
    IF Verbose>0 THEN Verb_=1
END IF
!
Printlen=FNDde_defprintlen ! define printlen for this session
!
! define colors and other settings
INTEGER Custom_colours
LONG Lctype
DIM Lcdata$(255)

```

```

!
Custom_colours=1                ! use custom colours (set to 0 if you do not wish to change colour palette)
!
IF NOT FNDde_init_system(Custom_colours) THEN STOP
!
IF Verb_>0 THEN CALL Dde_programhead("HTBasic - Office Interface Framework example #1")
!
!
! define the application (uncomment the desired application and comment out the remaining App$ strings!)
App$="Excel"      ! MS Excel
!App$="Word"      ! MS Word
!App$="Access"    ! MS Access
!App$="QPW"       ! Corel Quattro Pro
!App$="WPWIN"     ! Corel WordPerfect. IMPORTANT !!! You must select the proper version for Wpwinvers$ !!
!
! IMPORTANT !!! You must select the proper version number for Wpwinvers$ !!
! Valid numbers are: "7", "8", "9", "10", "11", "12", "13"
! Wpwinvers$="12" ! use this if you are using WordPerfect 12
! Wpwinvers$="13" ! use this if you are using WordPerfect X3 (release 13)
!
!
! Application quit behaviour
! Values:  0 = Do not close the application ever
!          1 = Close the application after all DDE commands have been sent without any warning
!          2 = Prompt the user to close the application
!          >2 = App_quit=Timeout value (prompt the user to close the application until the given timeout has reached)
!             If timeout occurs the dialog will be closed and the application will not be closed automatically
App_quit=2
!
IF UPC$(App$)<>"WPWIN" OR UPC$(App$)<>"QPW" THEN
!
! Mso2007=0 ! set this to 1 if you are using MS Office 2007
!
Office2007: Mso2007=-1 ! comment out this line if you don't want to ask for MS Office 2007
!
IF Mso2007=-1 THEN ! now ask if MS Office 2007 or above is being used
Question$="Are you using Microsoft Office 2007 ?"&Crlf$&Crlf$
Question$=Question$&"** Attention **"&Crlf$
Question$=Question$&" Select ""Yes"" only if you are really "&Crlf$
Question$=Question$&" using MS Office 2007, otherwise your "&Crlf$
Question$=Question$&" Office application may crash or may "&Crlf$
Question$=Question$&" become unstable !"
CALL Dde_yesno_widgt("MS Office 2007?",Question$,"Yes","No",Btn,1)
IF Btn=0 THEN
Mso2007=1! set this to a value>0 if you are using MS Office 2007 and above
ELSE
Mso2007=0! set this to a value>0 if you are using MS Office 2007 and above
END IF
END IF
!
IF Mso2007 THEN ! file type suffix for MS Office 2007 (and above) applications
Wordsuffix$=".docx"
Excelsuffix$=".xlsx"
Accesssuffix$=".accdb"
ELSE ! document file type suffixes for MS Office up to version 2003
Wordsuffix$=".doc"
Excelsuffix$=".xls"
Accesssuffix$=".mdb"
END IF
END IF
!
! *** Options for opening documents ***
! You can specify options when opening Microsoft Office documents (does not apply to Corel WordPerfect).
! Currently defined are:
! "RO" - Open the document in ReadOnly mode
!
Openopt$="RO" ! set this to "" if you want to open the document in normal mode
!
!
! ** ATTENTION **
! If you are using old 16 Bit MS Office programs you _must_ use the
! MS DOS compatible short form for directory strings. This does not apply
! to MS Office 97 and above !
!
! Example:
! Long directory: "C:\Program Files\Tech Soft\BASICDDE2\Examples\MS Office"
! Short form:    "C:\Progra~1\TechSo~1\BASICDDE2\Examples\MSoffi~1"
!
! If you do not use the short form your 16 Bit application may fail to
! load or save any document !
!
! You can use the library function FNDde_short_fname$ in order to get
! the short form of your directory and file name.
! Usage:
! Doc_template$(1)=FNDde_short_fname$("Long Directory path\Document_long_name.ext")
!
! Long Directory path = Long Directory path
! Document_long_name.ext = Long file name of desired document
!
! Again: This is not necessary for 32 Bit software like MS Office 97 and above or
! Corel WordPerfect Office !
!
!*****
!
! If the specified filename starts with a "*" a new file will be created and saved
! Newfile$="" ! uncomment this if you want to create a new file
!
IF Openopt$<>" " THEN Openopt$=":"&Openopt$ ! suffix ":"
!
SELECT UPC$(App$)
Doc_password$="" ! default: no write password
!
CASE "WORD"

```

```

Topic$="Winword"
Lctype=1
Lcdata$=FNDde_get_locale$(Lctype,Verb_) ! get country code from Windows version
!
IF Newfile$="*" THEN
! If the specified filename starts with a "*" a new file will be created and saved
Doc_template$(1)=Newfile$&&SYSTEM$("MSI")&"\newfile."&Wordsuffix$ ! Create new empty Word document
ELSE
Doc_template$(1)=SYSTEM$("MSI")&"\template."&Wordsuffix$ ! Word template without write password
!
! if we detect a United States Windows version we use a different template
IF Lcdata$="0409" THEN Doc_template$(1)=SYSTEM$("MSI")&"\template-letter."&Wordsuffix$ ! US version with "letter"
END IF
!
! Doc_password$="passwd" ! write password for Word documents
Doc_template$(2)=SYSTEM$("MSI")&"\my_document."&Wordsuffix$ ! Word filename for created document
!
CASE "EXCEL"
! some definitions for documents which should be automatically loaded into the application
! according to the selected application (other applications can be easily added)
Topic$="Excel"
Doc_password$="" ! write password for Word documents
!
IF Newfile$="*" THEN
! If the specified filename starts with a "*" a new file will be created and saved
Doc_template$(1)=Newfile$&&SYSTEM$("MSI")&"\newfile."&Excelsuffix$ ! Create new empty document
ELSE
Doc_template$(1)=SYSTEM$("MSI")&"\template."&Excelsuffix$ ! Open existing Excel template
END IF
Doc_template$(2)=SYSTEM$("MSI")&"\template-output."&Excelsuffix$ ! Excel output template
!
CASE "ACCESS"
Topic$="Msaccess"
! Caution !!
! Please select the proper example database *before* you start the skeleton program with MS Access
Doc_template$(1)=SYSTEM$("MSI")&"\template_2007."&Accesssuffix$ ! MS Access 2007 template / database
Doc_template$(1)=SYSTEM$("MSI")&"\template_xp."&Accesssuffix$ ! MS Access XP template / database
! Doc_template$(1)=SYSTEM$("MSI")&"\template_97."&Accesssuffix$ ! MS Access 97 template / database
! Doc_template$(1)=SYSTEM$("MSI")&"\templ_20."&Accesssuffix$ ! MS Access 2.0 (16 Bit) template / database
!
Doc_template$(2)=SYSTEM$("MSI")&"\template-output."&Accesssuffix$ ! MS Access output template
!
CASE "QPW"
Topic$="QPW"
!
IF Newfile$="*" THEN
! If the specified filename starts with a "*" a new file will be created and saved
Doc_template$(1)=Newfile$&&SYSTEM$("MSI")&"\newfile.qpw" ! Create new empty document
ELSE
Doc_template$(1)=SYSTEM$("MSI")&"\template.qpw" ! Quattro Pro template
END IF
Doc_template$(2)=SYSTEM$("MSI")&"\template-output.qpw" ! Quattro Pro output file
!
CASE "WPWIN"
Topic$="WPWIN"&Wpwinvers$
!
IF Newfile$="*" THEN
! If the specified filename starts with a "*" a new file will be created and saved
Doc_template$(1)=Newfile$&&SYSTEM$("MSI")&"\newfile.wpd" ! Create new empty document
ELSE
Doc_template$(1)=SYSTEM$("MSI")&"\template.wpd" ! WordPerfect template
END IF
Doc_template$(2)=SYSTEM$("MSI")&"\template-output.wpd" ! WordPerfect output file
!
CASE ELSE
BEEP
PRINT
ALPHA PEN 2
PRINT " ** WARNING: Unkown application string: App$='&App$&' "
ALPHA PEN 1
PRINT " Known values: Excel, Winword, Msaccess, QPW, WPWIN*"
PRINT
PRINT " Program will be continued anyway."
PRINT
END SELECT
!
IF Topic$<>" " AND App$<>" " THEN
App_exe$=Topic$&".exe"
ELSE
Topic$=App$
App_exe$=""
END IF
!
G$=CHR$(19) ! character code for inverse printing
X$=CHR$(16) ! character code for normal printing
H1$=CHR$(30) ! character code for "highlighted" printing (different color)
Hx$=CHR$(24) ! character code for "normal" printing (color)
!
CALL Dde_start(App_exe$,Topic$,Sys_chn(1),Language,Verb_) ! start the DDE application defined in Topic$
!
IF NOT Sys_chn(1) THEN
BEEP
GOTO Finish
END IF
!
ALPHA PEN 3
IF Verb_>0 THEN OUTPUT CRT;" "&CHR$(20)&"Sending DDE commands to "&Topic$&X$
ALPHA PEN 1
!
!
! first open a document in desired application (according to Topic$ and Doc_template(1)$)
Doc_chn(1)=FNDde_openfile(Topic$&Openopt$,Sys_chn(1),Doc_template$(1),Result(1),Doc_password$,Verb_)

```

```

!
IF NOT Doc_chn(1) THEN
  BEEP
  GOTO Finish
END IF
!
!
Myown_dde_cmds: !
!
! =====
! ***** Custom DDE programming starts here *****
!
! Todo: Write your own MS Office programming code here
CALL Dde_driver_info      ! shows info about DDE driver version and date
!
! ***** Custom DDE programming ends here *****
!
! =====
!
CALL Dde_focus2basic      ! bring HTBasic window back to front
!
!
IF Newfile$="*" THEN
  ! if the file was newly created simply save it under the given name
  Dummy=FNDde_savefile(Topic$,Doc_chn(1),Doc_template$(1),Result(1),2,"",Verb_)
ELSE
  ! Save file under the name specified in Doc_template$(2). Ask for overwrites
  Dummy=FNDde_savefile(Topic$,Doc_chn(1),Doc_template$(2),Result(1),0,"",Verb_)
END IF
!
Dde_enter_kbd(" Press <RETURN> to close the document(s).")
!
! Close file and ask if file should be saved
! Doc_chn(1)=FNDde_closefile(Topic$,Sys_chn(1),Doc_template$(1),Result(1))
!
! Close file, do not ask for save
Dummy=FNDde_closefile(Topic$,Sys_chn(1),Doc_template$(1+(1-Dummy)),Result(1),1,Verb_)
!
Finish: !
! terminate all DDE connections and resources
! if variable "App_quit" is set to 1 or greater the application will be closed
CALL Dde_quit(Sys_chn(1),Doc_chn(1),Result(1),App_quit,Topic$,Verb_)
!
ALPHA PEN 4
DISP " Program finished."
ALPHA PEN 1
!
SUBEXIT
!
! Global error handler
Dde_err_handler: !
BEEP
Dde_enter_kbd("**** Global Error Handler ****&CrLf$&CrLf$&ERRM$)
ERROR RETURN
!
SUBEND

```